Spherical Pendulum 06-09-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[11]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

I work through some calculations relevant to the spherical pendulum.

Included are

1 Derivation of equations of motion for the pendulum from the Lagrangian perspective.
2 Numerical solution of equations of motion and animation
3 Tractable special case - the conical pendulum
4 Perturbing the conical pendulum leads to precession
5 Comparing the approximate and numerical solutions

Of course there is lots more that could be added to this notebook. If I had the time, I would next work through the derivations of precession in *The Precessing Spherical Pendulum*, Olsson 1978, 1980. Another interesting addition would be to work through the amplitude equation approach in *Robert Hooke’s conical pendulum from the modern viewpoint of amplitude equations and its optical analogues, Rousseax,Coullet, Gilli 2006.*



Derivation of equations of motion

There many readily available sources of information on the spherical pendulum. For example

- http://www.cmi.ac.in/~souvik/books/mech/%20Tong.pdf
- Classical Mechanics, Corbin and Stehles

https://books.google.ie/books?id=1gxk4oq9trYC&pg=PA103&lpg=PA103&dq=%22classical+mechanics%22+%22spherical+pendulum%22&source=bl&ots=RvO9Qtu1wy&sig=kZWP75swVf-pbve8dm6IK8BD-Cj4&hl=en&sa=X&ved=0ahUKEwju54uQ-vXMAhXoL8AKHRZpBHoQ6AEIGzAA#v=onepage&q=%22-classical%20mechanics%22%20%22spherical%20pendulum%22&f=false

- http://rspa.royalsocietypublishing.org/content/462/2066/531

Derivation of equations of motion

I introduce the position of the pendulum bob.

```
In[13]:= w1[1] =
  {x[t] == L Sin[θ[t]] Cos[ϕ[t]], y[t] == L Sin[θ[t]] Sin[ϕ[t]], z[t] == -L Cos[θ[t]]}

Out[13]= {x[t] == L Cos[ϕ[t]] Sin[θ[t]], y[t] == L Sin[θ[t]] Sin[ϕ[t]], z[t] == -L Cos[θ[t]]}
```

Kinetic energy

In[14]:=
```
w1[2] = 𝒯[t] == m/2 (D[x[t], t]^2 + D[y[t], t]^2 + D[z[t], t]^2)
```

Out[14]=
$$\mathcal{T}[t] == \frac{1}{2} m \left(x'[t]^2 + y'[t]^2 + z'[t]^2\right)$$

Potential energy

In[15]:=
```
w1[3] = 𝒱[t] == m g z[t]
```

Out[15]=
$$\mathcal{V}[t] == g\, m\, z[t]$$

Lagrangian

In[16]:=
```
w1[4] = ℒ[t] == 𝒯[t] - 𝒱[t]
```

Out[16]=
$$\mathcal{L}[t] == \mathcal{T}[t] - \mathcal{V}[t]$$

In[17]:=
```
w1[5] = w1[4] /. RuleFromDefinition[w1[2]] /. RuleFromDefinition[w1[3]]
```

Out[17]=
$$\mathcal{L}[t] == -g\, m\, z[t] + \frac{1}{2} m \left(x'[t]^2 + y'[t]^2 + z'[t]^2\right)$$

In[18]:=
```
w1[6] = w1[5] /. (RuleFromDefinition /@ w1[1])
```

Out[18]=
$$\mathcal{L}[t] == g\, L\, m\, \mathrm{Cos}[\theta[t]] + \frac{1}{2} m \left(L^2 \mathrm{Sin}[\theta[t]]^2 \theta'[t]^2 + \left(L\, \mathrm{Cos}[\theta[t]]\, \mathrm{Sin}[\phi[t]]\, \theta'[t] + L\, \mathrm{Cos}[\phi[t]]\, \mathrm{Sin}[\theta[t]]\, \phi'[t]\right)^2 + \left(L\, \mathrm{Cos}[\theta[t]]\, \mathrm{Cos}[\phi[t]]\, \theta'[t] - L\, \mathrm{Sin}[\theta[t]]\, \mathrm{Sin}[\phi[t]]\, \phi'[t]\right)^2\right)$$

In[21]:=
```
w1[6] = MapEqn[Simplify, w1[6]]
```

Out[21]=
$$\mathcal{L}[t] == \frac{1}{2} L\, m \left(2\, g\, \mathrm{Cos}[\theta[t]] + L\, \theta'[t]^2 + L\, \mathrm{Sin}[\theta[t]]^2\, \phi'[t]^2\right)$$

Note that the Lagrangian does not explicitly depend on ϕ(t)

In[3]:=
```
Clear[RuleFromDefinition];
RuleFromDefinition[f_[v_] == rhs_] := f → Function[{v}, rhs]
```

The Euler Lagrange equation for θ is

In[30]:=
```
w1[7] = ((D[D[#, D[θ[t], t]], t] - D[#, θ[t]]) & /@ Expand@w1[6][[2]]) == 0 // Simplify
```

Out[30]=
$$L\, m \left(g\, \mathrm{Sin}[\theta[t]] - L\, \mathrm{Cos}[\theta[t]]\, \mathrm{Sin}[\theta[t]]\, \phi'[t]^2 + L\, \theta''[t]\right) == 0$$

In[31]:=
```
w1[8] = (#/(m L^2)) &/@ w1[7];
w1[8] = MapEqn[Simplify, w1[8]]
```

Out[32]=
$$\frac{g\,\mathrm{Sin}[\theta[t]]}{L} - \mathrm{Cos}[\theta[t]]\,\mathrm{Sin}[\theta[t]]\,\phi'[t]^2 + \theta''[t] == 0$$

Since the Lagrangian is independent of $\phi[t]$, the angular momentum in the ϕ direction is a constant of motion.

In[33]:=
```
def[κ] = κ == D[w1[6][[2]], D[ϕ[t], t]] // Simplify
```

Out[33]=
$$\kappa == L^2\,m\,\mathrm{Sin}[\theta[t]]^2\,\phi'[t]$$

In[34]:=
```
w1[9] = Simplify /@ (w1[8] /. Sol[def[κ], ϕ'[t]])
```

Out[34]=
$$-\frac{\kappa^2\,\mathrm{Cot}[\theta[t]]\,\mathrm{Csc}[\theta[t]]^2}{L^4\,m^2} + \frac{g\,\mathrm{Sin}[\theta[t]]}{L} + \theta''[t] == 0$$

This can be reduced to a one-parameter system by introducing a scale factor for the time. The oscillation frequency of a planar pendulum is a convenient choice.

In[35]:=
```
def[ω2] = ω^2 == g / L
```

Out[35]=
$$\omega^2 == \frac{g}{L}$$

Then

In[36]:=
```
w1[10] = w1[9] /. Sol[def[ω2], g]
```

Out[36]=
$$-\frac{\kappa^2\,\mathrm{Cot}[\theta[t]]\,\mathrm{Csc}[\theta[t]]^2}{L^4\,m^2} + \omega^2\,\mathrm{Sin}[\theta[t]] + \theta''[t] == 0$$

Define a dimensionless time variable

In[37]:=
```
def[τ] = τ == ω t
```

Out[37]=
$$\tau == t\,\omega$$

In[39]:=
```
w1[11] = w1[10] /. θ → ((θ[ω #]) &) /. Sol[def[τ], t]
```

Out[39]=
$$-\frac{\kappa^2\,\mathrm{Cot}[\theta[\tau]]\,\mathrm{Csc}[\theta[\tau]]^2}{L^4\,m^2} + \omega^2\,\mathrm{Sin}[\theta[\tau]] + \omega^2\,\theta''[\tau] == 0$$

ExpandAll

In[40]:=
```
w1[12] = MapEqn[(#/ω^2) &, w1[11]] // ExpandAll
```
Out[40]=
$$-\frac{\kappa^2 \, \text{Cot}[\theta[\tau]] \, \text{Csc}[\theta[\tau]]^2}{L^4 \, m^2 \, \omega^2} + \text{Sin}[\theta[\tau]] + \theta''[\tau] == 0$$

The initial angular momentum is expressed in terms of a single parameter.

In[41]:=
```
def[α] = α == κ^2/(L^4 m^2 ω^2)
```
Out[41]=
$$\alpha == \frac{\kappa^2}{L^4 \, m^2 \, \omega^2}$$

So

In[43]:=
```
w1[13] = w1[12] /. Sol[def[α], κ]
```
Out[43]=
$$-\alpha \, \text{Cot}[\theta[\tau]] \, \text{Csc}[\theta[\tau]]^2 + \text{Sin}[\theta[\tau]] + \theta''[\tau] == 0$$

For ϕ

In[44]:=
```
w1[14] = def[κ] /. Sol[def[α], κ];
w1[14] = MapEqn[(#/(L^2 m ω)) &, w1[14]]
```
Out[45]=
$$-\sqrt{\alpha} == \frac{\text{Sin}[\theta[t]]^2 \, \phi'[t]}{\omega}$$

In[46]:=
```
w1[15] = Solve[w1[14], ϕ'[t]][[1, 1]]
```
Out[46]=
$$\phi'[t] \to -\sqrt{\alpha} \, \omega \, \text{Csc}[\theta[t]]^2$$

or

In[49]:=
```
w1[16] = w1[15] /. {θ → ((θ[ω #]) &), ϕ → ((ϕ[ω #]) &)} /. Sol[def[τ], t] // RE;
w1[16] = MapEqn[(# / ω) &, w1[16]] // StandardizeEqn
```
Out[50]=
$$\sqrt{\alpha} \, \text{Csc}[\theta[\tau]]^2 + \phi'[\tau] == 0$$

For convenience, I collect the dimensionless form of the equations of motion

In[51]:=
```
w1["final"] =
  {-α Cot[θ[τ]] Csc[θ[τ]]^2 + Sin[θ[τ]] + θ''[τ] == 0, √α Csc[θ[τ]]^2 + ϕ'[τ] == 0}
```
Out[51]=
$$\left\{-\alpha \, \text{Cot}[\theta[\tau]] \, \text{Csc}[\theta[\tau]]^2 + \text{Sin}[\theta[\tau]] + \theta''[\tau] == 0, \ \sqrt{\alpha} \, \text{Csc}[\theta[\tau]]^2 + \phi'[\tau] == 0\right\}$$

In[52]:=
```
w1["final"] // ColumnForm
```

Out[52]=

$$-\alpha \,\text{Cot}[\theta[\tau]]\,\text{Csc}[\theta[\tau]]^2 + \text{Sin}[\theta[\tau]] + \theta''[\tau] == 0$$
$$\sqrt{\alpha}\,\text{Csc}[\theta[\tau]]^2 + \phi'[\tau] == 0$$

2 Numerical solution and animation

Before proceeding with analysis it is useful gain some perspective by generating a numerical solution



The θ motion is oscillatory as the pendulum rotates primarily in the -ϕ direction. An animation of the pendulum motion is instructive



3 Tractable special case - the conical pendulum

Analytical progress can be made for the special case of a conical pendulum, when the bob rotates in ϕ with θ constant.



The conical pendulum is important in the history of physics. It was the main topic of a series of letters exchanged by Robert Hooke and Isaac Newton circa 1680. These letters played a role in motivating Newton to renew his work in "natural philosophy" and hence contributed to his later work on gravity. The conical pendulum was also a factor in the famous dispute between these two great scientists. This history is discussed in the introduction of the article *Robert Hooke' s conical pendulum from the modern viewpoint of amplitude equations and its optical analogs*, G. Rousseaux, P. Coullet, J - M. Gilli, Proceedings of the Royal Society, Vol 462, Issue 2066, February 2006.
http : // rspa.royalsocietypublishing.org/content/462/2066/531

The rotation frequency of a conical pendulum can be determined from force balance

In[53]:=
```
w3[1] = {T Cos[θ0] == m g, T Sin[θ0] == m vϕ^2/r}
```
Out[53]= $\{T \cos[\theta_0] == g\, m,\ T \sin[\theta_0] == \frac{m\, v_\phi^2}{r}\}$

In[54]:=
```
w3[2] = w3[1][[2]] /. Sol[w3[1][[1]], T]
```
Out[54]= $g\, m \tan[\theta_0] == \frac{m\, v_\phi^2}{r}$

In[55]:=
```
w3[3] = Solve[w3[2] /. vϕ → r D[ϕ[t], t] /. r → L Sin[θ0], ϕ'[t]][[2, 1]] // RE
```
Out[55]= $\phi'[t] == \frac{\sqrt{g}\sqrt{\sec[\theta_0]}}{\sqrt{L}}$

In[56]:=
```
w3[4] = DSolve[{w3[3], ϕ[0] == 0}, ϕ[t], t][[1, 1]] // RE
```
Out[56]= $\phi[t] == \frac{\sqrt{g}\ t \sqrt{\sec[\theta_0]}}{\sqrt{L}}$

In[57]:=
```
w3[5] = Sol[w3[4] /. t → Tc /. ϕ[Tc] → 2 π /. Tc → 2 π / ωc, ωc] // RE
```
Out[57]= $\omega_c == \frac{\sqrt{g}\sqrt{\sec[\theta_0]}}{\sqrt{L}}$

This makes sense. When θ_0 is small, the frequency of the conical pendulum approaches that of a planar pendulum, $\sqrt{g/L}$. When θ_0 near $\pi/2$, the conical pendulum must rotate very rapidly to overcome the relatively stronger force of gravity.

The conical pendulum can also be analyzed in terms of the equations of motion derived in section 1. It is instructive to write this in analogy to an equation describing the motion of a particle in a potential.

In[58]:=
```
w3[6] = Solve[w1["final"], θ''[τ]][[1, 1]] // RE // ExpandAll
```
Out[58]=
$$\theta''[\tau] == \alpha\,\mathrm{Cot}[\theta[\tau]]\,\mathrm{Csc}[\theta[\tau]]^2 - \mathrm{Sin}[\theta[\tau]]$$

Introduce a potential D[V[θ], θ]

In[59]:=
```
w3[7] = -D[V[θ], θ] == (w3[6][[2]] /. θ[τ] → θ)
```
Out[59]=
$$-V'[\theta] == \alpha\,\mathrm{Cot}[\theta]\,\mathrm{Csc}[\theta]^2 - \mathrm{Sin}[\theta]$$

In[60]:=
```
w3[8] = MapEqn[Integrate[-#, θ] &, w3[7]]
```
Out[60]=
$$V[\theta] == -\mathrm{Cos}[\theta] + \frac{1}{2}\alpha\,\mathrm{Cot}[\theta]^2$$

This potential forms a well and the minimum of that well occurs at θ_0, the characteristic displacement angle for a conical pendulum. I illustrate the dependence of θ_0 on the parameter α.

In[61]:=
```
Module[{F, θMin, Lab},
 F[θ_, α_] := -Cos[θ] + 1/2 α Cot[θ]^2;
 θMin[α_] := FindMinimum[F[θ, α], {θ, π / 4}][[2, 1, 2]] ;
 Lab[α_] := StringForm["α = `` θmin = ``π", α, NF2[θMin[α] / π]];
 Plot[{F[θ, 0.1], F[θ, 0.5], F[θ, 1.0]},
  {θ, 0, π/2}, AxesLabel → {Stl["θ"], Stl["V(θ)"]},
  PlotLabel → Stl["Potential well for θ motion"],
  Ticks → {{0, π / 4, π/2, 3 π / 4}, Automatic}, PlotStyle → {Black, Blue, Green},
  PlotLegends → Placed[{Lab[0.1], Lab[0.5], Lab[1.0]}, Right]]]
```
Out[61]=


4 Perturbing the conical pendulum leads to precession

I want to make use of subscripts and load the Notation package.

In[62]:=
```
<< Notation`;
```

In[63]:=
```
Symbolize[ θ0 ]; Symbolize[ θ̇0 ];
Symbolize[ ϕ̇0 ];
Symbolize[ Ωδθ ];
Symbolize[ δθ0 ]
```

The general equations are

In[64]:=
```
w4[1] = w1["final"]
```

Out[64]=

$$\left\{-\alpha \operatorname{Cot}[\theta[\tau]] \operatorname{Csc}[\theta[\tau]]^2+\operatorname{Sin}[\theta[\tau]]+\theta''[\tau]==0,\ \sqrt{\alpha}\operatorname{Csc}[\theta[\tau]]^2+\phi'[\tau]==0\right\}$$

I impose perturbations

In[65]:=
```
w4[2] = w4[1] /. {θ → ((θ0 + ϵ δθ[#]) &) , ϕ′[τ] → ϕ̇0[τ] + ϵ δϕ′[τ]}
```

Out[65]=

$$\left\{-\alpha \operatorname{Cot}[\theta_0+\epsilon\,\delta\theta[\tau]] \operatorname{Csc}[\theta_0+\epsilon\,\delta\theta[\tau]]^2+\operatorname{Sin}[\theta_0+\epsilon\,\delta\theta[\tau]]+\epsilon\,\delta\theta''[\tau]==0,\ \sqrt{\alpha}\operatorname{Csc}[\theta_0+\epsilon\,\delta\theta[\tau]]^2+\dot\phi_0[\tau]+\epsilon\,\delta\phi'[\tau]==0\right\}$$

First consider the perturbations of the θ equation.

In[66]:=
```
w4[3] = MapEqn[Normal@Series[#, {ϵ, 0, 2}] &, w4[2][[1]] // ExpandAll
```

Out[66]=

$$-\alpha \operatorname{Cot}[\theta_0] \operatorname{Csc}[\theta_0]^2+\operatorname{Sin}[\theta_0]+\epsilon\operatorname{Cos}[\theta_0]\,\delta\theta[\tau]+\alpha\,\epsilon\operatorname{Csc}[\theta_0]^2\,\delta\theta[\tau]+3\,\alpha\,\epsilon\operatorname{Cot}[\theta_0]^2\operatorname{Csc}[\theta_0]^2\,\delta\theta[\tau]-4\,\alpha\,\epsilon^2\operatorname{Cot}[\theta_0]\operatorname{Csc}[\theta_0]^2\,\delta\theta[\tau]^2-6\,\alpha\,\epsilon^2\operatorname{Cot}[\theta_0]^3\operatorname{Csc}[\theta_0]^2\,\delta\theta[\tau]^2-\frac{1}{2}\epsilon^2\operatorname{Sin}[\theta_0]\,\delta\theta[\tau]^2+\epsilon\,\delta\theta''[\tau]==0$$

The lowest order equation relates the angular momentum parameter α to θ_0

In[67]:=
```
w4[4] = Sol[w4[3] /. ϵ → 0, α]
```

Out[67]=

$$\alpha\to\operatorname{Sin}[\theta_0]^3\operatorname{Tan}[\theta_0]$$

The first order equation yields

In[68]:=
```
w4[5] = w4[3] /. ϵ^2 → 0 /. ϵ → 1 /. w4[4] // StandardizeEqn;
w4[5] = MapEqn[Collect[#, δθ[τ]] &, w4[5]]
```

Out[69]=

$$\left(4\operatorname{Cos}[\theta_0]+\operatorname{Sin}[\theta_0]\operatorname{Tan}[\theta_0]\right)\delta\theta[\tau]+\delta\theta''[\tau]==0$$

I define a frequency of convenience

In[70]:=
```
def[Ω_δθ] = Ω_δθ^2 == (4 Cos[θ_0] + Sin[θ_0] Tan[θ_0])
```

Out[70]= $\Omega_{\delta\theta}^2 == 4\,\text{Cos}[\theta_0] + \text{Sin}[\theta_0]\,\text{Tan}[\theta_0]$

In[71]:=
```
w4[6] = w4[5] /. (Reverse[def[Ω_δθ]] // ER)
```

Out[71]= $\Omega_{\delta\theta}^2\,\delta\theta[\tau] + \delta\theta''[\tau] == 0$

Solve this for initial conditions $\delta\theta(0) = \delta\theta_0$, $\delta\theta'(0) = 0$. The pendulum bob is displaced from the conical equilibrium value by an amount $\delta\theta_0$ and released at rest.

In[72]:=
```
w4[7] = DSolve[{w4[6], δθ[0] == δθ0, δθ'[0] == 0}, δθ[τ], τ][[1, 1]]
```

Out[72]= $\delta\theta[\tau] \to \delta\theta0\,\text{Cos}[\tau\,\Omega_{\delta\theta}]$

Next consider the equation for ϕ

In[73]:=
```
w4[8] = w4[2][[2]]
```

Out[73]= $\sqrt{\alpha}\,\text{Csc}[\theta_0 + \epsilon\,\delta\theta[\tau]]^2 + \dot{\phi}_0[\tau] + \epsilon\,\delta\phi'[\tau] == 0$

In[74]:=
```
w4[9] = MapEqn[Normal@Series[#, {ϵ, 0, 2}] &, w4[8]] // ExpandAll
```

Out[74]= $\sqrt{\alpha}\,\text{Csc}[\theta_0]^2 - 2\sqrt{\alpha}\,\epsilon\,\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]^2\,\delta\theta[\tau] + \sqrt{\alpha}\,\epsilon^2\,\text{Csc}[\theta_0]^2\,\delta\theta[\tau]^2 + 3\sqrt{\alpha}\,\epsilon^2\,\text{Cot}[\theta_0]^2\,\text{Csc}[\theta_0]^2\,\delta\theta[\tau]^2 + \dot{\phi}_0[\tau] + \epsilon\,\delta\phi'[\tau] == 0$

The lowest order equation is

In[75]:=
```
w4[10] = w4[9] /. ϵ → 0 /. w4[4] // Simplify[#, Assumptions → {0 < θ_0 < π/2}] &
```

Out[75]= $\sqrt{\text{Sec}[\theta_0]} + \dot{\phi}_0[\tau] == 0$

which may be integrated

In[76]:=
```
w4[11] = DSolve[{w4[10] /. φ̇_0[τ] → D[φ_0[τ], τ] , φ_0[0] == 0}, φ_0[τ], τ][[1, 1]]
```

Out[76]= $\phi_0[\tau] \to -\tau\sqrt{\text{Sec}[\theta_0]}$

The first order equation is

In[77]:=
```
w4[12] =
  w4[9] /. ϵ^2 → 0 /. ϵ → 1 /. w4[4] // Simplify[#, Assumptions → {0 < θ_0 < π/2}] &
```

Out[77]= $\sqrt{\text{Sec}[\theta_0]} + \dot{\phi}_0[\tau] + \delta\phi'[\tau] == 2\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\,\delta\theta[\tau]$

Remove the dependence on $\dot{\phi}_0[\tau]$

In[78]:=
```
w4[13] = w4[12] /. Sol[w4[10], φ̇0[τ]] // Simplify
```

Out[78]=
$$2\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\;\delta\theta[\tau] == \delta\phi'[\tau]$$

Introduce the $\delta\theta$ dependence

In[79]:=
```
w4[14] = w4[13] /. w4[7]
```

Out[79]=
$$2\,\delta\theta 0\,\text{Cos}[\tau\,\Omega_{\delta\theta}]\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]} == \delta\phi'[\tau]$$

In[80]:=
```
w4[15] = DSolve[{w4[14], δϕ[0] == 0}, δϕ[τ], τ][[1, 1]]
```

Out[80]=
$$\delta\phi[\tau] \to \frac{1}{\Omega_{\delta\theta}} 2\,\delta\theta 0\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\;\text{Sin}[\tau\,\Omega_{\delta\theta}]$$

Summarizing, the perturbation solutions are

In[81]:=
```
w4[16] = θ[τ] == θ0 + δθ[τ] /. w4[7]
```

Out[81]=
$$\theta[\tau] == \theta_0 + \delta\theta 0\,\text{Cos}[\tau\,\Omega_{\delta\theta}]$$

In[82]:=
```
w4[17] = ϕ[τ] == ϕ0[τ] + δϕ[τ] /. w4[11] /. w4[15]
```

Out[82]=
$$\phi[\tau] == -\tau\sqrt{\text{Sec}[\theta_0]} + \frac{1}{\Omega_{\delta\theta}} 2\,\delta\theta 0\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\;\text{Sin}[\tau\,\Omega_{\delta\theta}]$$

In[83]:=
```
w4["final"] =
  {θ[τ] == θ0 + δθ0 Cos[τ Ωδθ], ϕ[τ] == -τ Sqrt[Sec[θ0]] + 1/Ωδθ 2 δθ0 Sqrt[Cot[θ0] Csc[θ0]] Sin[τ Ωδθ]}
```

Out[83]=
$$\left\{\theta[\tau] == \theta_0 + \delta\theta 0\,\text{Cos}[\tau\,\Omega_{\delta\theta}],\ \phi[\tau] == -\tau\sqrt{\text{Sec}[\theta_0]} + \frac{1}{\Omega_{\delta\theta}} 2\,\delta\theta 0\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\;\text{Sin}[\tau\,\Omega_{\delta\theta}]\right\}$$

In[84]:=
```
w4["final"] // ColumnForm
```

Out[84]=
$$\theta[\tau] == \theta_0 + \delta\theta 0\,\text{Cos}[\tau\,\Omega_{\delta\theta}]$$
$$\phi[\tau] == -\tau\sqrt{\text{Sec}[\theta_0]} + \frac{2\,\delta\theta 0\sqrt{\text{Cot}[\theta_0]\,\text{Csc}[\theta_0]}\;\text{Sin}[\tau\,\Omega_{\delta\theta}]}{\Omega_{\delta\theta}}$$

5 Comparing the approximate and numerical solutions

It is sometimes awkward to work with subscripts inside functions

In[85]:=
```
w4["final"] /. Sol[def[Ω_δθ], Ω_δθ] /. θ_0 → θ0
```

Out[85]=

$$\{\theta[\tau] == \theta0 + \delta\theta0\,\mathrm{Cos}\left[\tau\sqrt{(4\,\mathrm{Cos}[\theta0] + \mathrm{Sin}[\theta0]\,\mathrm{Tan}[\theta0])}\right],\ \phi[\tau] == -\tau\sqrt{\mathrm{Sec}[\theta0]} + \left(2\,\delta\theta0\sqrt{\mathrm{Cot}[\theta0]\,\mathrm{Csc}[\theta0]}\,\mathrm{Sin}\left[\tau\sqrt{(4\,\mathrm{Cos}[\theta0] + \mathrm{Sin}[\theta0]\,\mathrm{Tan}[\theta0])}\right]\right) \Big/ \left(\sqrt{(4\,\mathrm{Cos}[\theta0] + \mathrm{Sin}[\theta0]\,\mathrm{Tan}[\theta0])}\right)\}$$

Define functions for the expressions for θ and ϕ

In[86]:=
```
Clear[Ωfcn, θApprox, ϕApprox];
Ωfcn[θ0_] := √(4 Cos[θ0] + Sin[θ0] Tan[θ0]);
θApprox[τ_, θ0_, δθ0_] := θ0 + δθ0 Cos[τ √(4 Cos[θ0] + Sin[θ0] Tan[θ0])];
ϕApprox[τ_, θ0_, δθ0_] :=
  -τ √Sec[θ0] + (2 δθ0 √(Cot[θ0] Csc[θ0]) Sin[τ √(4 Cos[θ0] + Sin[θ0] Tan[θ0])]) /
    (√(4 Cos[θ0] + Sin[θ0] Tan[θ0]))
```

Polar representation of the θ–ϕ motion
θ is the 'radial' coordinate and ϕ is the 'angular' coordinate
numerical (black) perturbation theory (dashed black)



The black point denotes the starting position at $\theta(\tau = 0) = \theta_0 + \delta\theta0$, $\phi(0) = 0$. In first order the pendulum rotates in the ϕ direction while oscillating back and forth between θ_0 - $\delta\theta0$ and θ_0 + $\delta\theta0$. The red numbers indicate the position of the pendulum bob at the first three times the θ excursion is at a maximum. Notice that the value of ϕ is not the same at these times, as it would be if the orbit was an ellipse. Rather, the ϕ values at maximum θ excursion precess in the -ϕ direction.

The perturbation theory results are quite accurate.

In[90]:=

```
Module[{α = 0.1, τMax = 20, L = 1, θ0 = π / 4,
  ϕ0 = 0, soln, results, path, lab, frames, BobPosition},
 BobPosition[{t_, θt_, ϕt_}] := {L Sin[θt ] Cos[ϕt ], L Sin[θt ] Sin[ϕt], -L Cos[θt ]};
 soln = NDSolve[{θ''[τ] == α Cot[θ[τ]] Csc[θ[τ]]^2 - Sin[θ[τ]], ϕ'[τ] == -√α Csc[θ[τ]]^2,
    θ[0] == θ0, θ'[0] == 0, ϕ[0] == ϕ0}, {θ[τ], ϕ[τ]}, {τ, 0, τMax}];
 results = Table[Flatten[{τ, θ[τ] /. soln, ϕ[τ] /. soln}], {τ, 0, τMax, 0.2}];
 path = BobPosition /@ results;
 frames =
  Table[ShowPendulum[results〚i〛, path〚1 ;; i〛, {θ0, ϕ0, α}], {i, 1, Length[path]}];
 ListAnimate[frames, 2]]
```

Out[90]=



Graphics

Spherical pendulum

Initial diagram

In[92]:=
```
Clear[QθArcArrow3DLabeledV2, QϕArcArrow3DLabeledV2];
QθArcArrow3DLabeledV2[rArrow_, θS_, θF_, ϕ_, text_, rText_] :=
 {Arrowheads[Small], Arrow[Table[CoordinateTransform["Spherical" -> "Cartesian",
 {rArrow, θ, ϕ}], {θ, θS, θF, Sign[θF - θS] Pi / 64}]],
   {Black, Text[text, CoordinateTransform["Spherical" -> "Cartesian", {rText,
    (θS + θF) / 2, ϕ}]]}};
QϕArcArrow3DLabeledV2[rArrow_, ϕS_, ϕF_, θ_, text_, rText_] :=
 {Arrowheads[Small], Arrow[Table[CoordinateTransform["Spherical" -> "Cartesian",
 {rArrow, θ, ϕ}], {ϕ, ϕS, ϕF, Sign[ϕF - ϕS] Pi / 64}]],
  {Black, Text[text, CoordinateTransform["Spherical" -> "Cartesian", {rText,
   θ, (ϕS + ϕF) / 2}]]}}
```

In[95]:=

```
Module[{L = 1, θ0 = π / 4, ϕ0 = π / 4, δ = 0.05, scale = 1.0, style = Black,
  mult = 1.1, pendulum, axes, θarc, ϕarc, O, P, Q, OQP, g, θArc, ϕArc},
 (* relevant points *)
 O = {0, 0, 0};
 P = {L Sin[θ0] Cos[ϕ0], L Sin[θ0] Sin[ϕ0], -L Cos[θ0]};
 Q = {L Sin[θ0] Cos[ϕ0], L Sin[θ0] Sin[ϕ0], 0};
 OQP = {Directive[Black, Dashed], Line[{O, Q, P}]};
 (* displayed elements *)
 axes = Module[{Ox = scale {1, 0, 0}, Oy = scale {0, 1, 0}, Oz = scale {0, 0, -1}},
   {style, Line[{O, Ox}], Text["x", mult Ox], Line[{O, Oy}],
    Text["y", mult Oy], Line[{O, Oz}], Text["-z", mult Oz]}];
 pendulum = {Directive[Brown, Thickness[0.015]], Line[{O, P}],
   {Black, PointSize[0.03], Point[P]}};
 (* Note that the function that calculates θArc requires that the
   starting and ending θ values be expressed in standard spherical
   coordinates with θ measures with respect to the positive z-axis *)
 θArc = {Gray,
   With[{rArrow = 0.4, θSt = π - 0.001, θFn = 3 π / 4, ϕConst = π / 4, rText = 0.45},
    QθArcArrow3DLabeledV2[rArrow, θSt, θFn, ϕConst, Stl["θ"], rText]]};
 ϕArc = {Gray,
   With[{rArrow = 0.4, ϕSt = 0.001, ϕFn = π / 4, θConst = π/2, rText = 0.45},
    QϕArcArrow3DLabeledV2[rArrow, ϕSt, ϕFn, θConst, Stl["ϕ"], rText]]};
 g[0] = ParametricPlot3D[{{Cos[u] Sin[v], Sin[v] Sin[u], Cos[v]}},
   {u, 0, 2 π}, {v, π/2, π}, Mesh → False, ColorFunction → "TemperatureMap",
   PlotStyle → Opacity[.25], Axes → None, Boxed → False, ViewPoint → {4.5, 2.5, 1},
   ImagePadding → 20, PlotLabel → Stl["Spherical Pendulum"], ImageSize → 400];
 (* overlap the displayed elements *)
 g[1] = Graphics3D[
   {axes, pendulum, OQP, θArc, ϕArc}, Boxed → False, ImageSize → 400];
 Show[{g[0], g[1]}]]
```

Out[95]=



Visualization of pendulum used for animation

In[96]:=
```
Clear[ShowPendulum];
ShowPendulum[{t_, θt_, ϕt_}, path_, {θ0_, ϕ0_, α_}] :=
 Module[{L = 1, δ = 0.05, scale = 1.0,
   style = Black, mult = 1.1, pendulum, axes, refLine, O, P, lab, g},
  (* relevant points *)
  O = {0, 0, 0};
  P = {L Sin[θt] Cos[ϕt], L Sin[θt] Sin[ϕt], -L Cos[θt]};
  (* displayed elements *)
  axes = Module[{Ox = scale {1, 0, 0}, Oy = scale {0, 1, 0}, Oz = scale {0, 0, -1}},
    {style, Line[{O, Ox}], Text["x", mult Ox], Line[{O, Oy}],
     Text["y", mult Oy], Line[{O, Oz}], Text["-z", mult Oz]}];
  pendulum = {Directive[Brown, Thickness[0.015]], Line[{O, P}],
    {Black, PointSize[0.03], Point[P]}};
  refLine = {Directive[Black, Dashed], Line@
     Table[{L Sin[θ0] Cos[ϕ], L Sin[θ0] Sin[ϕ], -L Cos[θ0]}, {ϕ, 0, 2 π, π/16}]};

  lab = Stl@StringForm["Spherical pendulum α = ``  τ = ``", α, t];
  g[0] = ParametricPlot3D[{{Cos[u] Sin[v], Sin[v] Sin[u], Cos[v]}},
    {u, 0, 2 π}, {v, π/2, π}, Mesh → False, ColorFunction → "TemperatureMap",
    PlotStyle → Opacity[.25], Axes → None, Boxed → False, ViewPoint → {4.5, 2.5, 4},
    ImagePadding → 20, PlotLabel → lab, ImageSize → 400];
  (* overlap the displayed elements *)
  g[1] = Graphics3D[
    {axes, pendulum, Line[path], refLine}, Boxed → False, ImageSize → 400];
  Show[{g[0], g[1]}]]
```

Display Conical Pendulum

In[103]:=
```
Module[{t = 0, θ0 = π / 4, ϕ0 = π/2, α = 0.1},
 ShowConicalPendulum[{0, θ0, ϕ0}, {θ0, ϕ0, α}]]
```

Out[103]=



In[99]:=
```
Clear[QArrow];
QArrow[vList_] := Arrow[Tube[vList]];
```

In[101]:=
```
Clear[ShowConicalPendulum];
ShowConicalPendulum[{t_, θt_, ϕt_}, {θ0_, ϕ0_, α_}] :=
 Module[{L = 1, δ = 0.05, scale = 1.0, style = Black, mult = 1.1,
   pendulum, axes, refLine, O, P, mgVec, TVec, FcentrifugalVec, θArc, lab, g},
  (* relevant points *)
  O = {0, 0, 0};
  P = {L Sin[θt] Cos[ϕt], L Sin[θt] Sin[ϕt], -L Cos[θt]};
  (* displayed elements *)
  axes = Module[{Ox = scale {1, 0, 0}, Oy = scale {0, 1, 0}, Oz = scale {0, 0, -1}},
    {style, Line[{O, Ox}], Text["x", mult Ox], Line[{O, Oy}],
     Text["y", mult Oy], Line[{O, Oz}], Text["-z", mult Oz]}];
  pendulum = {Directive[Brown, Thickness[0.005]], Line[{O, P}],
    {Black, PointSize[0.03], Point[P]}};
  refLine = {Directive[Black, Dashed], Arrow@
     Table[{L Sin[θ0] Cos[ϕ], L Sin[θ0] Sin[ϕ], -L Cos[θ0]}, {ϕ, 2 π, 0, -π/16}]};

  mgVec = {Blue, QArrow[{P, P + {0, 0, -0.3 L}}],
    {Black, Text[Stl["mg⃗"], P + {0, 0, -0.35 L}]}};
  TVec = {Blue, QArrow[{P, 0.67 P }], {Black, Text[Stl["T⃗"], 0.67 P + {0, 0.15, 0}]}};
  FcentrifugalVec = {Blue, QArrow[{P, P + 0.3 {0, 1, 0} }],
    {Black, Text[Stl["F⃗c"], P + 0.35 {0, 1, 0}]}};
  θArc = {Gray,
    With[{rArrow = 0.4, θSt = π - 0.001, θFn = π - θ0, ϕConst = ϕ0, rText = 0.45},
     QθArcArrow3DLabeledV2[rArrow, θSt, θFn, ϕConst, Stl["θ0"], rText]]};

  lab = Stl@StringForm["Spherical pendulum α = ``  τ = ``", α, t];
  g[1] = Graphics3D[
    {axes, pendulum, refLine, mgVec, TVec, FcentrifugalVec, θArc},
    Boxed → False, ImageSize → 400]]
```